
ABSTRACT  

Occasionally, SAS programmers receive very wide SAS data sets with hundreds or thousands of 
variables.  The data could have been extracted from Blaise, downloaded from a SQL Server table, 
obtained from an Oracle table, or sent by a client.  Not all of the variables in the large SAS data set may 
have data stored in them.  This can happen for any number of reasons, such as partial surveys, test 
tables, database tables with variables in them for future use, and upstream processing issues. 
 
It is useful to know which variables actually have values stored in them. Analysts can concentrate on 
analyzing those variables and skip the others.  Completely empty variables may indicate a problem with a 
file.  Missing values might signify an issue with the collection instrument’s skip patterns.  Perhaps the 
chronically missing variables can be dropped from the SAS data set or from the originating database 
tables.  Additionally, it is sometimes helpful to know which variables hold values for the individual cases 
within the SAS data set. 
 
The SAS Data Set Missing Value Analyzer identifies all the variables that have one or more non-missing 
value in them for a given SAS data set.  The program creates four types of Excel report files: 
 

• Count of the Variables That Have Non-Missing Values For Each Distinct ID Variable  

• Characteristics of All Variables With At Least One Non-Missing Value 

• Characteristics of Variables With ALL Missing Values 

• Optionally, a separate report for each distinct ID Variable titled: Characteristics of The Variables 
with One or More Non-Missing Values 

 
This paper reviews the SAS Data Set Missing Value Analyzer macro program, and the report files it 
creates. Readers can begin using the macro directly after reading this paper. 

INTRODUCTION  

A SAS missing value is a value that indicates that no data value is stored for the variable in the current 
observation.  SAS has several types of missing values: 
 

• Numeric – Represented by a single decimal point: . 

• Character – Represented by “” 

• Special Missing – Numeric variables only that are represented by .a through .z and ._ 
 
Note that projects sometimes use negative numbers to signify missing data.  For instance, -7, -8, -9, and 
so on.  However, these are not true SAS missing values; and because of the breadth of possibilities, such 
missing values are not covered in this paper. 
 
Figure 1 provides an example of a SAS data set that has missing values. The values highlighted in 
yellow indicate missing values.  Notice that some variables (TOTINC, TOTINC_AT, VALUE, and WAGES) 
have low amounts of missingness, some variables (RETIR, and SEMPI) have high amounts of 
missingness, and some variables (NOTES, EMAIL, TOPCODE, and BOTTOMCODE) are completely 
missing for every observation.  

Figure 1. Graphical view of a SAS data set with missing values. 

 
The SAS Data Set Missing Value Analyzer (SDSMVA) offers an automated method of checking for 
missingness within a SAS data set by identifying all variables that have one or more non-missing values 
in them.  The SDSMVA creates four types of Excel report files: 
 

• Count of the variables that have non-missing values for each Distinct ID variable 

• Characteristics of all variables with at least one non-missing value 

• Characteristics of variables with all missing values 

• OPTIONALLY: A separate report for each distinct ID variable titled: Characteristics of the 
Variables with One or More Non-Missing Values 

 
Output 1 is an example of a report showing the number of variables that have data in them for each 
unique ID variable in a data set.  Only two observations have non-missing data for all 147 variables.  And 
one observation—with DealerID “DEL000002” has missing values for all but the DealerID variable. 

 
Output 1. Count of the variables that have non-missing values for each Distinct ID variable 
 
Output 2 is an example of a report showing the characteristics of all variables that have at least one non-
missing value in them. 

Output 2.  Characteristics of all variables with at least one non-missing value 
 
Output 3 is an example of a report of all variables that contain nothing but missing values for the entire 
SAS data set. 

 
Output 3. Characteristics of variables with all missing values 
 
Output 4 is an example of one of the reports that was generated for each of the hundreds of distinct 
DealerID’s in the SAS data set. That is; one report was generated for each observation in the SAS data 
set. It specifies the characteristics of each variable that does not contain missing variables for that 
particular observation.  Note that, since there is a potential for hundreds or even thousands of these 
reports to be generated, this is a purely optional report. 

Output 4. Report for the observation where DealerID equals “DEL00001”showing the 
characteristics all variables with non-missing values  
 
The MissingValueAnalyzer macro has five parameters: 
 

• DIRNAME - The directory of the target SAS data set 
• DSNAME - The name of the target SAS data set 
• IDVARIABLE - An ID variable that uniquely identifies an observation 
• REPTDIR - The directory where the output Excel report files will be written 
• INDIVIDUALREPORTS - A value of Y will produce a separate Excel report file for each 

distinct ID Variable 
 
Display 1 is an example of specifying the parameters for the MissingValueAnalyzer macro. 
 

 
Display 1. Specifying the parameters for the MissingValueAnalyzer macro. 
 
The next five sections provide the annotated code found in the MissingValueAnalyzer macro.  Note that 
the entire program can be found in Appendix A. 
 

PART 0 – OPTIONS AND FILE SPECIFICATIONS 

The program opens with the macro statement listing the parameters that will be passed to the macro. 
Several ODS statements perform various housekeeping tasks. A libref of InputLib is specified for the 
directory holding the target SAS data set.  The full path name of the directory is input via the &DIRNAME 
macro variable. See Display 2. 

Display 2.  
 

PART 1 – CREATING A REPORT OF THE NUMBER OF NON-MISSING-VALUE 
VARIABLES PER UNIQUE ID VARIABLE 

This section of the program starts with a DATA step that is going to determine the name, length, label, 
format, and type (numeric or character) of each variable in the target SAS data set.  The ID variable is 
passed to the KEEP option of the DATA statement via the &IDVARIABLE macro variable.  The name of 
the target data set is passed to the SET statement via the &DSNAME macro variable.  Next, two arrays 
are created: one containing all of the numeric variables, the other containing all of the character variables.  
A LABEL and LENGTH statement complete the first part of this DATA step. See Display 3. 
 

 
Display 3.  
 
Next, two DO loops are used to determine the name, length, label, and format of each variable. The first 
DO loop works on the numeric variables, using the numeric array previously created. It uses a group of 
variable functions (VNAME, VLENGTH, VLABEL, and VFORMAT) to identify each numeric variable’s 
metadata.  The second DO loop does the same thing for the character variables. 
 
Then the data are sorted by the ID variable. See Display 4 and Display 5. 

Display 4. 

 
Display 5. 
 
The program continues by running a frequency against the new, sorted data set (FILLEDVARS) and 
writing the frequency output to data set FREQOUT. Then ODS is used to create the first report 
spreadsheet, which is a count of all variables having non-missing values in the observation for each 
unique ID variable.  This is the report that was illustrated in Output 1.  See Display 6. 

Display 6. 
 

PART 2 – CREATING A REPORT WITH A LIST OF VARIABLES THAT HAD ONE OR 
MORE NON-MISSING VALUES IN THEM 

A simple PROC SORT is used to sort the FILLEDVARS data set into ascending alphabetical order by the 
VARNAME variable. Then, the report of all variables with one or more non-missing values in them is 
created. This is the report that was illustrated in Output 2.  See Display 7. 
 

 
Display 7. 
 

PART 3 – CREATING A REPORT OF CHARACTERISTICS OF VARIABLES WITH 
ALL MISSING VALUES 

This section of the program is the busies with a SORT, PROC DATASETS, another SORT, a MERGE, 
and a DATA _NULL_ before running a PROC PRINT to create a report of the characteristics of all 
variables that have at least one non-missing value in them in the target SAS data set. 
 
The sort obtains all variable attributes from the target SAS data set from sashelp.vcolumn, sorts them into 
variable name order, and outputs them to a temporary SAS data set named ALLVARS.  The DATASETS 

procedure is then used to rename variables in ALLVARS and to assign labels to them.  ALLVARS will be 
used in a future merge in this section.  See Display 8. 
 

 
Display 8. 
 
Part 3 continues by sorting FILLEDVARS by VARNAME for the upcoming match-merge.  Then, 
FILLEDVARS and ALLVARS are match-merged together into new SAS data set ALLMISSING. Only 
observations that are in ALLVARS but not in FILLEDVARS are kept in the new data set.  So, 
ALLMISSING will contain only variables that did not have a single value in them—the variables that had 
missing values in them throughout the entire SAS data set.  See Display 9. 
 

 
Display 9 
 
The last section of Part 3 first determines if there are any variables in the ALLMISSING data set.  If so, 
then the macro variable &NUMMISSING will be set to the value 1 or higher.  If not, then it will be zero.  

So, if the value of NUMMISSING is not equal to zero, then ALLMISSING is sorted by VARNAME and the 
report is created.  This is the report that was illustrated in Output 3. See Display 10. 
 

 
Display 10. 
 

PART 4 – CREATING A REPORT OF THE NUMBER OF NON-MISSING-VALUE 
VARIABLES FOR EACH INDIVIDUAL, UNIQUE ID VARIABLE 

This final section of the program is only run if the optional &INDIVIDUALREPORTS macro variable was 
set to “Y” by the person running the macro. If & INDIVIDUALREPORTS = Y, then a separate report file is 
created for each observation in the SAS data set.  The beginning of each report file’s name is the value of 
the ID variable (&IDVARIABLE) so that the reports have distinct names and can be correlated with the ID 
that uniquely identifies them. 
 
If invoked (&INDIVIDUALREPORTS = Y), this section begins by determining the number of observations 
in the data set and storing it in macro variable TOTOBS.  The next PROC SQL stores the value of each 
ID variable (identified by macro parameter &IDVARIABLE) into a macro array of variables CASEID1, 
CASEID2, CASEID3, and so on. See Display 11. 

Display 11. 
 
Now, a macro DO loop is executed to create each individual report using a DATA step and a PROC 
PRINT.  The macro DO loop executes from 1 to the value of &TOTOBS, which is the number of the last 
observation in the FILLEDVARS data set. The DATA step subsets FILLEDVARS to a single observation, 
based on the value stored in the CASEIDx macro variable.  The ODS statements use the current value of 
CASEIDx to create a unique Excel file for the report.  The Print procedure prints the report for the 
particular observation in the original SAS data set.  This is the report that was illustrated in Output 4. See 
Display 12. 
 
The %MEND statement signals the end of the %MissingValuesAnalyzer macro program. 

Display 12. 
 

OPERATIONALIZING THE SAS DATA SET MISSING VALUES ANALYZER 

You can operationalize the SAS Data Set Missing Values Analyzer.sas program by copying it from 
Appendix A into a new .sas program file on your computer with the same name.  Once the program has 
been copied, move it to one of your autocall macro libraries. This will ensure that when you invoke the 
program with %MissingValuesAnalyzer(and specifying the proper parameters), it will automatically be 
invoked. 
 
An alternative is to copy the driver program, Execute SAS Data Set Missing Values Analyzer.sas, from 
Appendix B into a new .sas program file on your computer.  Modify the %INCLUDE statement in that file 
to reflect where you stored the SAS Data Set Missing Values Analyzer.sas program.  Then, fill in the 
parameters and run the program.  The code below illustrates the %INCLUDE and the parameters that 
must be specified: 

 

 
Display 13. 

CONCLUSION 

In very large SAS data sets, it is useful to know which variables actually hold values.  Analysts can 
concentrate on those variables and skip the others.  Missing values may indicate a problem with the file, 
with a data collection instrument’s skip patterns, or with upstream processing.  You may be able to drop 
variables with all missing values altogether.  Additionally, it can be helpful to know which variables do 
have values for individual cases. 
 
The SAS Data Set Missing Value Analyzer identifies all the variables that have one or more non-missing 
value in them for a given SAS data set.  The program creates four types of Excel report files: 
 

• Count of the Variables That Have Non-Missing Values For Each Distinct ID Variable  

• Characteristics of All Variables With At Least One Non-Missing Value 

• Characteristics of Variables With ALL Missing Values 

• Optionally, a separate report for each distinct ID Variable titled: Characteristics of The Variables 
with One or More Non-Missing Values 

 
So, copy the code in Appendix A and B and start analyzing your own organization’s large SAS data sets 
for missing values. 

APPENDIX A. THE SAS DATA SET MISSING VALUES ANALYZER.SAS PROGRAM 

/***************************************************************************/ 

/*  Program: SAS Data Set Missing Values Analyzer.sas                      */ 

/*                                                                         */ 

/*  Author: Michael A. Raithel                                             */ 

/*                                                                         */ 

/*  Created: 04/05/2023                                                    */ 

/*                                                                         */ 

/*  Purpose: This program examines a SAS data set and identifies all       */  

/*  variables that have one or more non-missing values in them. It is      */ 

/*  designed to allow a relatively quick way for analysts to inspect very  */ 

/*  "wide" SAS data sets and skip over possibly thousands of variables with*/ 

/*  missing values to zero in on only those that actually contain data that*/  

/*  can be analyzed. The program creates four types of Excel spreadsheet   */ 

/*  reports--see "Outputs" for details.                                    */ 

/*                                                                         */ 

/*  Parameters: Users must specify the following parameters:               */ 

/*                                                                         */ 

/*             DIRNAME    - The directory of the target SAS data set       */ 

/*             DSNAME     - The name of the target SAS data set            */ 

/*             IDVARIABLE - An ID variable that uniquely identifies an     */ 

/*                          observation - REQUIRED!                        */  

/*             REPTDIR    - The directory where the output Excel report    */ 

/*                          files will be written                          */ 

/*             INDIVIDUALREPORTS - A value of "Y" (no quotes, please) will */ 

/*                                 produce a separate Excel report file for*/ 

/*                                 each distinct IDVARIABLE.               */ 

/*                                                                         */ 

/*  Outputs: This program creates the following Excel report files in the  */ 

/*  &REPTDIR directory:                                                    */ 

/*                                                                         */ 

/*           1."&DSNAME Count of Variables That Have Non-Missing Values    */ 

/*            For Each Distinct IDVARIABLE..xlsx" - A report with one row  */ 

/*            for each unique IDVARIABLE with the number of variables with */ 

/*            non-missing values in that particular observation.           */ 

/*                                                                         */                                                                                                    

/*          2."&DSNAME - Characteristics of All Variables With At Least    */ 

/*            One Non-Missing Value.xlsx" A data set level report of all   */ 

/*            variables that have at least one non-missing value in them.  */   

/*            it includes whether the variable is character or numeric,    */ 

/*            length, format, and label.                                   */  

/*             label.                                                      */                                 

/*                                                                         */                     

/*           3. "&DSNAME - Characteristics of Variables With ALL MISSING   */  

/*              VALUES" - This Excel report file is generated for variables*/   

/*              that have missing values for EVERY OBSERVATION in the      */ 

/*              target SAS data set. If all of the variables have at least */ 

/*              one non-missing value, the report file is not generated.   */ 

/*                                                                         */ 

/*           4. "&&caseid&i.. Variables with One or More Non-Missing       */ 

/*              Values.xlsx" - An individual Excel report file is generated*/ 

/*              for EACH unique IDVARIABLE listing the variables with non- */ 

/*              missing values in that specific observation. It includes   */ 

/*              whether the variable is character or numeric, and  the     */ 

/*              variable's length, format, and label.                      */ 

/*                                                                         */ 

/*              NOTE: MACRO PARAMETER INDIVIDUALREPORTS=Y MUST BE SPECIFIED*/  

/*              FOR THE INDIVIDUAL REPORTS TO BE PRODUCED! IT IS OKAY TO   */ 

/*              LEAVE IT BLANK OR SPECIFY: N IF YOU DO NOT WANT THE        */ 

/*              INDIVIDUAL-LEVEL REPORTS.                                  */  

/*                                                                         */ 

/***************************************************************************/ 

 

%MACRO MisssingValuesAnalyzer(DIRNAME=, DSNAME=, IDVARIABLE=, REPTDIR=,     

INDIVIDUALREPORTS=); 

 

/* Options and file specifications */ 

options nofmterr symbolgen mprint mlogic; 

 

ods html close; 

ods listing close; 

ods results off; 

ods noproctitle; 

 

libname InputLib "&DIRNAME" access=readonly; 

 

/***************************************************************************/ 

/*                                     PART 1                              */ 

/***************************************************************************/ 

/*                                                                         */ 

/* Create a report of the number of non-missing-value variables per unique */ 

/* &IDVARIABLE                                                             */ 

/*                                                                         */ 

/***************************************************************************/ 

 

/* Read target SAS data set and only keep variables */ 

/* that DO NOT have missing values in them.         */ 

data filledvars(keep= &IDVARIABLE varname vartype varlength varlabel  

                      varformat); 

set  InputLib.&DSNAME; 

 

array nums{*}  _numeric_; 

array chars{*} _character_; 

 

label varname   = "Variable Name" 

 varlength = "Length" 

 varlabel  = "Label" 

 varformat = "Format" 

      vartype   = "Numeric or Character" 

   ; 

length varname   $32. 

       varlength 8. 

  varlabel  $256. 

  varformat $49. 

       vartype   $4. 

    ; 

 

 

/* Inspect numeric variables for non-missing values and keep the */ 

/* names, lengths, labels, and formats of non-missing variables  */ 

do I = 1 to dim(nums); 

 

 if not missing(nums{i}) then do; 

 

  varname   = vname(nums{i}); 

  varlength = vlength(nums{i}); 

  varlabel  = vlabel(nums{i}); 

  varformat = vformat(nums{i}); 

 

  vartype = "num"; 

 

  output; 

 

 end; 

end; 

 

/* Inspect character variables for non-missing values and keep the */ 

/* names, lengths, labels, and formats of non-missing variables    */ 

do I = 1 to dim(chars); 

 

 if not missing(chars{i}) then do; 

 

  varname = vname(chars{i}); 

  varlength = vlength(chars{i}); 

  varlabel  = vlabel(chars{i}); 

  varformat = vformat(chars{i}); 

 

  vartype = "char"; 

  

  output; 

 

 end; 

end; 

 

run; 

 

/*Sort the data set by &IDVARIABLE varname to get data in &IDVARIABLE order*/ 

proc sort data=filledvars; 

 by &IDVARIABLE varname; 

run; 

 

/* Calculate the frequency of non-missings for each individual &IDVARIABLE */ 

proc freq data=filledvars noprint; 

 tables &IDVARIABLE / out=freqout norow nocol nopercent nocum ; 

 title1 "Number of Variables That Have Non-Missing Values"; 

 title2 "For Each Distinct: &IDVARIABLE"; 

run; 

 

/* Create the spreadsheet report of the number of non-missing-value */  

/* variables by &IDVARIABLE                                         */ 

 

ODS EXCEL file="&REPTDIR\&DSNAME Count of Variables That Have Non-Missing  

                Values For Each Distinct &IDVARIABLE..xlsx"; 

 

ODS EXCEL options(sheet_name="Obs With Nonmissing Values"  

                  flow="header,data"); 

 

proc print noobs data=freqout label; 

 var &IDVARIABLE count; 

title1 "Count of Variable Variables That Have Non-Missing Values"; 

title2 "For Each Distinct: &IDVARIABLE"; 

label count = "Number of Variables with Non-missing Values"; 

run; 

 

ODS EXCEL CLOSE; 

 

 

/***************************************************************************/ 

/*                                     PART 2                              */ 

/***************************************************************************/ 

/*                                                                         */ 

/* Create a spreadsheet with a list of variables that have one or more     */ 

/* values across ALL of the observations in the specified SAS data set.    */ 

/*                                                                         */ 

/***************************************************************************/ 

 

/* Sort the data into variable name order */ 

proc sort nodupkey data=filledvars(keep=varname vartype varlength varformat  

                                        varlabel) out=uniquevars; 

 by varname; 

run; 

 

/* Create the Excel report file in the specified report directory */ 

ODS EXCEL file="&REPTDIR\&DSNAME - Characteristics of All Variables With At  

                Least One Non-Missing Value.xlsx"; 

 

ODS EXCEL options(sheet_name="Vars With Nonmissing Values"  

                  flow="header,data"); 

 

proc print noobs data = uniquevars label; 

var varname vartype varlength varformat varlabel; 

title1 "Characteristics of All Variables With At Least One Non-Missing 

Value"; 

run; 

 

ods EXCEL close; 

 

/***************************************************************************/ 

/*                                     PART 3                              */ 

/***************************************************************************/ 

/*                                                                         */ 

/* Create a spreadsheet with a list of variables that have ONLY MISSING    */  

/* VALUES across ALL of the observations in the specified SAS data set.    */ 

/*                                                                         */ 

/***************************************************************************/ 

 

/* Sort all of the target data set variable metadata into variable name */ 

/* order                                                                */ 

proc sort data=sashelp.vcolumn(where=(libname="INPUTLIB" and  

                                      memname=upcase("&DSNAME")))  

      out=allvars(keep=name type length label format); 

 by name; 

run; 

 

/* Rename and label selected variables for the upcoming report */ 

proc datasets library=work noprint; 

 modify allvars; 

 rename name   = varname 

        length = varlength 

   label  = varlabel 

   format = varformat 

   type   = vartype 

   ; 

 

 label varname   = "Variable Name" 

       varlength = "Length" 

       varlabel  = "Label" 

       varformat = "Format" 

  vartype   = "Variable Type" 

       ; 

 

run; 

quit; 

 

/* Sort previously-created filledvars into variable name order for the match-

merge with allvars */ 

proc sort data=filledvars; 

 by varname; 

run; 

 

/* Match merge with filledvars to find only variables with _ALL_ missing 

values */ 

data allmissing; 

merge filledvars(in=a) 

 allvars(in=b) 

      ; 

by varname; 

 

if b and not a; 

 

run; 

 

/* Determine if there are any observations in ALLMISSING */ 

data _null_; 

set  sashelp.vtable(where=(libname = "WORK" and memname = "ALLMISSING" and  

                           memtype = "DATA")); 

 

call symput("NUMMISSING", nobs); 

 

run; 



18 

 

/* Only create this report if there are observations in ALLMISSING */ 

%IF &NUMMISSING NE 0 %THEN %DO; 

 

 /* Sort the allmissing variables by varname for printing */ 

 proc sort data=allmissing sortseq=linguistic(numeric_collation=on); 

  by varname; 

 run; 

 

 /* Create the Excel report file in the specified report directory */ 

 ODS EXCEL file="&REPTDIR\&DSNAME - Characteristics of Variables With  

                      ALL MISSING VALUES.xlsx"; 

 

 ODS EXCEL options(sheet_name="Vars With ALL MISSING Values"  

                        flow="header,data"); 

 

 proc print noobs data = allmissing label; 

  var varname vartype varlength varformat varlabel; 

 title1 "Characteristics of Variables With All Missing Values"; 

 run; 

 

 ods EXCEL close; 

 

%END; 

 

/***************************************************************************/ 

/*                                     PART 4                              */ 

/***************************************************************************/ 

/*                                                                         */ 

/* Create a report of the number of non-missing-value variables for each   */ 

/* individual, unique value  of the &IDVARIABLE variable. This results in  */ 

/* as many spreadsheets as there are &IDVARIABLEs. Each spreadsheet name   */ 

/* will begin with the value of its &IDVARIABLE. NOTE: This section is only*/ 

/* executed if the &INDIVIDUALREPORTS parameter was coded: Y               */ 

/*                                                                         */ 

/***************************************************************************/ 

 

%IF %UPCASE(&INDIVIDUALREPORTS) = Y %THEN %DO; 

 

***********************************************************; 

* Determine number of observations in filledvars data set *; 

***********************************************************; 

proc sql noprint; 

create table &IDVARIABLE as 

 select distinct &IDVARIABLE 

  from filledvars; 

 

%let TOTOBS = &SQLOBS; 

 

quit; 

 

***********************************************************; 

* Load &IDVARIABLE values into distinct macro variables   *; 

***********************************************************; 

proc sql noprint; 

select distinct &IDVARIABLE into : caseid1 - :caseid&TOTOBS 

       from filledvars; 

quit; 

 

**********************************************************************; 

* Loop to create the individual SAS data sets and Excel spreadsheets *; 

**********************************************************************; 

%Do i = 1 %to &totobs %by 1; 

 

 /* Subset the data set for each individual value of IDVARIABLE */ 

 data Individual; 

 set  filledvars(where=(&IDVARIABLE = "&&caseid&i")); 

 run; 

 

 /* Create the Excel report file in the specified report directory */ 

 ODS EXCEL file="&REPTDIR\&&caseid&i.. Variables with One or More Non- 

                      Missing Values.xlsx"; 

 

 ODS EXCEL options(sheet_name="&&caseid&i.. Non-missing Vars"  

                        flow="header,data"); 

 

 proc print noobs data = Individual label; 

  var varname vartype varlength varformat varlabel; 

 title1 "Variables with One or More Non-Missing Values for &IDVARIABLE:  

              &&caseid&i "; 

 

 run; 

 

 ods EXCEL close; 

 

%END; 

 

%END; 

 

 

%MEND MisssingValuesAnalyzer; 

 

APPENDIX B. THE EXECUTE SAS DATA SET MISSING VALUES ANALYZER.SAS 
DRIVER PROGRAM 

/***************************************************************************/ 

/*  Program: Execute SAS Data Set Missing Values Analyzer.sas              */ 

/*                                                                         */ 

/*  Author: Michael A. Raithel                                             */ 

/*                                                                         */ 

/*  Created: 04/05/2023                                                    */ 

/*                                                                         */ 

/*  Purpose: This is the driver program for the SAS Data Set Missing Values*/ 

/*  Analyzer macro program.                                                */ 

/*                                                                         */ 

/*  That SAS macro program examines a SAS data set and identifies all      */  

/*  variables that have one or more non-missing values in them. It is      */ 

/*  designed to allow a relatively quick way for analysts to inspect very  */ 

/*  "wide" SAS data sets and skip over possibly thousands of variables with*/ 

/*  missing values to zero in on only those that actually contain data that*/  

/*  can be analyzed. The program creates four types of Excel spreadsheet   */ 

/*  reports--see "Outputs" for details.                                    */ 

/*                                                                         */ 

/*  Parameters: Users must specify the following parameters:               */ 

/*                                                                         */ 

/*             DIRNAME    - The directory of the target SAS data set       */ 

/*             DSNAME     - The name of the target SAS data set            */ 

/*             IDVARIABLE - An ID variable that uniquely identifies an     */ 

/*                          observation - REQUIRED!                        */  

/*             REPTDIR    - The directory where the output Excel report    */ 

/*                          files will be written                          */ 

/*             INDIVIDUALREPORTS - A value of "Y" (no quotes, please) will */ 

/*                                 produce a separate Excel report file for*/ 

/*                                 each distinct IDVARIABLE.               */ 

/*                                                                         */ 

/*  Outputs: This program creates the following Excel report files in the  */ 

/*  &REPTDIR directory:                                                    */ 

/*                                                                         */ 

/*           1."&DSNAME Count of Variables That Have Non-Missing Values    */ 

/*            For Each Distinct IDVARIABLE..xlsx" - A report with one row  */ 

/*            for each unique IDVARIABLE with the number of variables with */ 

/*            non-missing values in that particular observation.           */ 

/*                                                                         */                                                                                                    

/*          2."&DSNAME - Characteristics of All Variables With At Least    */ 

/*            One Non-Missing Value.xlsx" A data set level report of all   */ 

/*            variables that have at least one non-missing value in them.  */   

/*            it includes whether the variable is character or numeric,    */ 

/*            length, format, and label.                                   */  

/*             label.                                                      */                                 

/*                                                                         */                     

/*           3. "&DSNAME - Characteristics of Variables With ALL MISSING   */  

/*              VALUES" - This Excel report file is generated for variables*/   

/*              that have missing values for EVERY OBSERVATION in the      */ 

/*              target SAS data set. If all of the variables have at least */ 

/*              one non-missing value, the report file is not generated.   */       

/*                                                                         */ 

/*           4. "&&caseid&i.. Variables with One or More Non-Missing       */ 

/*              Values.xlsx" - An individual Excel report file is generted */ 

/*              for EACH unique IDVARIABLE listing the variables with non- */ 

/*              missing values in that specific observation. It includes   */ 

/*              whether the variable is character or numeric, and  the     */ 

/*              variable's length, format, and label.                      */ 

/*                                                                         */ 

/*              NOTE: MACRO PARAMETER INDIVIDUALREPORTS=Y MUST BE SPECIFIED*/  

/*              FOR THE INDIVIDUAL REPORTS TO BE PRODUCED! IT IS OKAY TO   */ 

/*              LEAVE IT BLANK OR SPECIFY: N IF YOU DO NOT WANT THE        */ 

/*              INDIVIDUAL-LEVEL REPORTS.                                  */  

/*                                                                         */ 

/***************************************************************************/ 

 

%INCLUDE "<YOUR PROJECT DRIVE>\SAS Data Set Missing Values Analyzer.sas"; 

 

%MisssingValuesAnalyzer(DIRNAME=,  

    DSNAME=,  

    IDVARIABLE=,  

    REPTDIR=, 

    INDIVIDUALREPORTS=);  


